NUCLEAR INSTRUMENTS AND METHODS 62 (1968) 205–207; © NORTH-HOLLAND PUBLISHING CO.

A COMPUTER PROGRAM FOR THE CALCULATION OF TANDEM ACCELERATOR BEAM TUBE OPTICS*

J. G. CRAMER

Department of Physics, University of Washington, Seattle, Washington, U.S.A.

Received 25 January 1968

A FORTAN II computer program for the calculation of particle trajectories through inclined-field or straight accelerating tubes of the type commonly used in Tandem Van de Graaff accelerators is described. Details of the program and results of tube optics calculations are discussed.

Accelerator tubes are one of the most expensive and vulnerable parts of a tandem accelerator. In an accelerator using "straight" tubes, i.e., tubes with electrode surface perpendicular to the direction of acceleration, a faulty acceleration gap can be shorted out without greatly impairing the operation of the accelerator. However, with the introduction of inclined-field tubes[1]) by High Voltage Engineering Corporation, the shorting of a single accelerator gap can radically alter the trajectory of particles through the accelerator and seriously damage the performance of the machine. Moreover, many considerations of accelerator development and improvement require some detailed information on the optics of the accelerator. Such information is not, unfortunately, provided by the manufacturer. For these reasons, an elementary computer program has been written in FORTRAN II for the computation of particle trajectories through inclined (or straight) acceleration tubes, as well as other optical elements in an accelerator system.

In the writing of this program, it was first necessary to derive the transfer equations which apply to electrostatic inclined-field accelerating tubes. This was accomplished by considering the non-relativistic equations for a non-inclined acceleration gap, and then performing a rotation of the coordinate system from the field axis to the tube axis. The transfer equations given below are the result of this derivation.

The equations for the position and inclination $(x_1, \dot{x}_1)$ of a single charged particle emerging from an accelerating gap, in terms of the entrance position and inclination $(x_0, \dot{x}_0)$, the potentials V_0 and V_1 of the entrance and exit acceleration electrodes, the length L of the gap, and the inclination angle θ of the electric field in the accelerating gap, are:

$$x_1 = x_0 + L[\{2\dot{x}'_0/(1+R)\} + \operatorname{tang}\theta],$$

$$\dot{x}_1 = (\dot{x}'_0 \cos\theta + R\sin\theta)/(R\cos\theta - \dot{x}'_0 \sin\theta),$$

where

$$\dot{x}'_0 = (\dot{x}_0 \cos\theta - \sin\theta)/(\dot{x}_0 \sin\theta + \cos\theta) =$$

$$= \text{particle inclination with respect to field,}$$

$$R = (V_1/V_0)^{\frac{1}{2}}.$$

Here we have used the convention that all potentials are referred to a zero potential at which the charged particle has zero velocity. Thus R above represents the ratio of exit to entrance velocity. Unfortunately, these equations are non-linear, so that the usual matrix methods[2]) can not be used. This non-linearity does not present any serious problems in the calculation, as examination of the program will show.

Several approximations were used in preparing the computer program, and they are enumerated here. The first was the assumption of non-relativistic kinematics. The principal justification for this assumption is that the uncertainties in tube and resistor construction introduce errors of such a magnitude that the refinement of relativistic mechanics is not warranted. A second assumption was that all magnetic fields due to electron-suppression magnets, etc., could be neglected. A third assumption was that the particle in passing through each acceleration gap covers a distance along the tube axis equal to the axial gap width. This is not quite true because inclined field gaps have slanting boundaries, so that if a particle enters low and exits high it may traverse more (or less) than one gap width between boundaries. However, errors due to this assumption will tend to average out. A fourth assumption is that of the aperture lens formula given by eq. (5-20) of Livingston and Blewett[3]). This formula is applied to describe the lens action which occurs when a particle passes from one gap to another and the two gaps have different electric fields. Finally, a gap which is a transition element between a straight section and an inclined field section is treated as having half the inclination of the inclined field section.

* Work was supported in part by the U.S. Atomic Energy Commission.

```
!JOB:
!ASSIGN S=MT0; SI=CR1W; BI=CR1W, BO=CP1W; LO=LP, X1=MT1W .
!FORTRAN SI,LO;BO.
 C       TANDEM BEAM-TUBE OPTICS CALCULATION
 C
 C       V(1) IS THE INITIAL ENERGY OF THE ACCELERATED IONS
 C       XD(1) IS THE INITIAL ANGLE OF THE BEAM WITH THE Z AXIS
 C       DV IS THE NORMAL VOLTAGE RISE PER SECTION
 C       X(1)IS THE INITIAL POSITION OF THE BEAM
 C       XSC AND XDSC ARE SCALE FACTORS USED IN PLOTTING OUT THE RESULTS
 C         (THESE ARE SET EQUAL TO 1.0 IF LEFT BLANK.)
 C       IELEM IS THE TYPE OF TUBE ELEMENT
 C             =0 FOR END OF TUBE
 C             =1 FOR INCLINED FIELD SECTION
 C               P1=GAP WIDTH PER SECTION
 C               P2=VOLTAGE RISE PER SECTION/DV
 C               P3=INCLINATION ANGLE
 C             =2 FOR DRIFT TUBE SECTION
 C               P1=LENGTH PER SECTION
 C             =3 FOR STEERER
 C               P1=DEFLECTION ANGLE(RADIANS) PER SECTION
 C             =4 FOR LENS
 C               P1=FOCAL LENGTH OF LENS(CM) PER SECTION
 C             (MAY BE EXPANDED FOR OTHER ELEMENTS LATER)
 C       NELEM IS THE NUMBER OF IDENTICAL ELEMENTS IN A PARTICULAR TUBE SEC
 C
 C
         DIMENSION V(300),X(300),XD(300),ANG(300)
         DIMENSION IX(300),IXD(300)
         EQUIVALENCE (X,IX),(XD,IXD)
     100 READ 110,V(1),DV,X(1),XD(1),XSC,XDSC
     110 FORMAT(7E10.5)
     111 IF(XSC) 112,112,113
     112 XSC=1.0
     113 IF(XDSC) 114,114,115
     114 XDSC=1.0
     115 N1=1
     116 E1=0.
     120 READ 130,IELEM,NELEM,P1,P2,P3,P4,P5
     130 FORMAT(2I5,5E10.5)
     140 DO 800 N=N1+1,N1+NELEM
     145 N2=N
     150 GO TO (820,160,300,330,360),IELEM+1
 C       INCLINED FIELD TUBE SECTION
     160 ANGLE=P3*0.0174533
     161 STH=SIN(ANGLE)
     162 CTH=COS(ANGLE)
     163 E2=P2*DV/(P1*CTH)
     175 V(N)=P2*DV+V(N-1)
     180 FR=(E2-E1)/(2.*(V(N-1)+V(N)))
     190 A=XD(N-1)-X(N-1)*FR
     200 ANG(N)=P3
     210 E1=E2
     220 R21=SQRT(V(N)/V(N-1))
     250 B=(A*CTH-STH)/(A*STH+CTH)
     260 C=R21+1.
     270 X(N)=X(N-1)+P1*(2.*B/C+STH/CTH)
     280 XD(N)=(B*CTH+R21*STH)/(R21*CTH-B*STH)
     290 GO TO 800
 C       DRIFT TUBE SECTION
     300 X(N)=X(N-1)+XD(N-1)*P1
     310 XD(N)=XD(N-1)
     315 ANG(N)=0.
     316 V(N)=V(N-1)
     320 GO TO 800
 C       STEERER SECTION
     330 X(N)=X(N-1)
     340 XD(N)=XD(N-1)+P1
     345 ANG(N)=P1
     346 V(N)=V(N-1)
     350 GO TO 800
 C       LENS SECTION
     360 X(N)=X(N-1)
     370 XD(N)=XD(N-1)-X(N-1)/P1
     375 ANG(N)=0.
     376 V(N)=V(N-1)
     380 GO TO 800
 C
 C       END OF TUBE CALCULATION
     800 CONTINUE
     805 N1=N2
     810 GO TO 120
     820 IF(SENSE SWITCH 4) 860,822
     822 PRINT 825
     825 FORMAT(1H125X*TANDEM ACCELERATOR BEAM OPTICS CALCULATION*///)
     830 PRINT 835
     835 FORMAT(6X*ELEMENT*8X*ANGLE*14X*ENERGY*14X*POSITION*13X*SLOPE*/20X*
        1DEGREES*15X*MEV*16X*CM*15X*RADIANS*//)
     840 PRINT 850,(N,ANG(N),V(N),X(N),XD(N),N=1,N1)
     850 FORMAT(I10,1P4E20.6)
     860 DO 890 N=1,N1
     870 IX(N)=X(N)*XSC*124.0+125.5
     880 IXD(N)=XD(N)*XDSC*2480.0+125.5
     890 CONTINUE
     900 PRINT 825
     910 CALL HIPLOT(IX,IXD,N1,125,20,25,0.,20.)
     920 GO TO 100
         END

PROGRAM ALLOCATION

  00016 X          00017 IX         01146 XD         01147 IXD
  02276 V          03426 ANG        04556 N1         04557 IELEM
  04560 NELEM      04561 N          04562 N2         04563 DV
  04565 XSC        04567 XDSC       04571 E1         04573 P1
  04575 P2         04577 P3         04601 P4         04603 P5
  04605 ANGLE      04607 STH        04611 CTH        04613 E2
  04615 FR         04617 A          04621 R21        04623 B
  04625 C

SUBPROGRAMS REQUIRED

  SIN         COS         SQRT        HIPLOT
THE END
```

Fig. 1. A compilation listing of the FORTRAN II program is shown. (Equal marks and sequential numbers are generated by the compiler and are not a part of the card format.)

Fig. 1 shows a compilation listing of the FORTRAN program, as compiled on the SDS 930 on-line computer system of this laboratory. The program is written in standard FORTRAN II and should run on any computer, except for one incompatibility. The subroutine HIPLOT called by this program is a plotting routine which uses the special high-density plotting symbols present on our line printer. In adapting the program to another computer, another plotting routine (or dummy subroutine) should be substituted.

Fig. 2 shows a typical particle trajectory through the University of Washington FN tandem accelerator, as calculated by this program. The beam is injected at 60 keV along the tube axis, and the accelerator has a terminal potential of 5.46 MV, so that the energy of the emerging particles is 10.98 MeV. Both the position and the angle of the beam are plotted. It is interesting to note that although the beam is injected on-axis, it emerges parallel and 2 mm above the tube axis.

Many exploratory calculations have been made with this code, studying such things as the effect of resistor variations on accelerator optics, the effect produced by shorting inclined field sections and methods of compensation, and the possible improvements in accelerator optics produced by terminal steering and focusing[4]). However, since these calculations are directly applicable only to our accelerator, they will not be discussed here.

One rather interesting result of such calculations is perhaps worth mentioning; in tracing rays backward from the stripper to the low energy end of the machine, it was discovered that a real image of the stripper is formed by these rays. If the beam entering the accelerator passes through this stripper-image, it will pass through the stripper itself and presumably through the whole accelerator, provided it does not collide with an accelerating electrode or alter its trajectory by small-angle scattering with residual gas in the tube. Fig. 3 shows this stripper-image for a terminal potential of 5.46 MeV and 60 keV injected protons. The position of the stripper-image depends, of course, on the terminal potential of the accelerator, and the injection energy of the accelerated particles, but the image position shown in fig. 3 is fairly typical and provides a clear idea of the requirements on the ion-source optics to achieve maximum transmission through the accelerator.

The author wishes to thank Prof. F. H. Schmidt and Mr. J. S. Heagney for helpful discussions while this work was in progress, and to acknowledge helpful correspondence with Dr. J. D. Larson of Brookhaven National Laboratory, who has been making similar



Fig. 2. Typical particle trajectory through the University of Washington FN tandem accelerator, as calculated by the program described here. Beam is injected on-axis at 60 keV. The terminal potential is 5.46 MV.



Fig. 3. Image of stripper formed just in front of entrance to low energy tube, assuming 60 keV protons injected, and a 5.46 MV terminal potential.

calculations and was kind enough to send some of his results for comparison.

References

[1] K. H. Purser, A. Galejs, P. H. Rose, R. J. Van de Graaff and A. B. Wittkower, Rev. Sci. Instr. **36** (1965) 453.

[2] S. Penner, Rev. Sci. Instr. **32** (1961) 150.

[3] M. S. Livingston and J. P. Blewett, *Particle accelerators* (McGraw-Hill, New York, 1962).

[4] E. Preikschat and J. G. Cramer, Rev. Sci. Instr. **37** (1966) 1722.